Gleichungssystem mit drei Unbekannten

Dokumentnummer: DX1649
Fachgebiet: Lineare Algebra, Gleichungssysteme, Matrizenrechnung, Determinanten
Einsatz: 2HAK (erstes Lernjahr)
Didaktischer Hinweis: Technologieeinsatz ist unbedingt erforderlich!

Figure 1:

Programmbeilage	
Diese Programme müssen installiert sein:	
CAS Maxima von http://maxima.sourceforge.net (in der Windows-Installationsdatei ist wxMaxima integriert)	Und GUI wxMaxima http://wxmaxima.sourceforge.net

1 Aufgabe

Figure 2: Gleichungssysteme sind nicht schwierig zu lösen, aber ohne Technologieeinsatz sehr aufwändig.

Man löse die folgenden Gleichungssysteme

a) mit dem Eliminationsverfahren
b) mit der Matrizenmultiplikation
c) mit der Determinantenmethode

```
(%i46) kill(all);
(%o0) done
```

2 Grundbeispiel

2.1 Eliminationsverfahren

Mit dem Eliminationsverfahren werden lineare Gleichungsysteme von 3 Gleichungen mit 3 Unbekannten sehr häufig gelöst. Das geht auch ohne Computereinsatz noch einigermaßen, was man bei größeren Gleichungssystemen nicht mehr behaupten kann.

Figure 3:

Die Matrix A lautete:	Der Vektor b lautete:	Die Lösung lautet:
-2, 1, -3, 7, -3, 10, 6, -2, 7.	2, 0, 0.	-2, 22, 8.

```
(%i1) g1:-2*x+y-3*z=2;
      g2:7*x-3*y+10*z=0;
      g3:6*x-2*y+7*z=0;
```

(%o1) $-3z+y-2x=2$

(%o2) $10z-3y+7x=0$

(%o3) $7z-2y+6x=0$

CAS-Lösung
(so geht es ganz schnell)

```
(%i4) l:solve([g1,g2,g3],[x,y,z]);
```

(%o4) $[[x=-2,y=22,z=8]]$

```
(%i5) l:algsys([g1,g2,g3],[x,y,z]);
```

(%o5) $[[x=-2,y=22,z=8]]$

Für das Eliminationsverfahren kann auch
Maxima nur als Rechenhilfe eingesetzt werden!

Elimination von z

```
(%i6) G1:10*g1+3*g2,expand
      /* erste und zweite Gleichung */;
      G2:7*g2-10*g3,expand
      /* zweite und dritte Gleichung */;
```

(%o6) $y+x=20$

(%o7) $-y-11x=0$

Elimination von y

```
(%i8) G:G1+G2;
```

(%o8) $-10x=20$

Lösung für x

```
(%i9) G:G/(-10);
```

(%o9) $x=-2$

Lösung für y
(durch Rückeinsetzen von x)

```
(%i10) G2:G2,G;
```

(%o10) $22-y=0$

```
(%i11) G2:G2-22;
```

(%o11) $-y=-22$

```
(%i12) G2:-G2;
```

(%o12) $y=22$

Lösung für z
(durch Rückeinsetzen von x und y)

```
(%i13) g1:g1,G,G2;
```

(%o13) $26-3z=2$

```
(%i14) g1:g1-26;
(%o14) -3 z=-24
```

```
(%i15) g1:g1/(-3);
(%o15) z=8
```

```
(%i16) Loesungsmenge:[G,G2,g1];
(%o16) [x=-2,y=22,z=8]
```

2.2 Matrizenmethode

Figure 4:

Die Matrix A lautete:
-2, 1, -3,
7, -3, 10,
6, -2, 7.

```
(%i17) A: matrix(
        [-2,1,-3],
        [7,-3,10],
        [6,-2,7]
       );
```

(%o17) $\begin{bmatrix} -2 & 1 & -3 \\ 7 & -3 & 10 \\ 6 & -2 & 7 \end{bmatrix}$

Figure 5:

Der Vektor b lautete:
2,
0,
0.

```
(%i18) b: matrix(
        [2],
        [0],
        [0]
       );
```

(%o18) $\begin{bmatrix} 2 \\ 0 \\ 0 \end{bmatrix}$

Figure 6:

Die Lösung lautet:
-2,
22,
8.

Es gibt zwei Berechnungsmöglichkeiten für die inverse Matrix. Eine Verwendung der Matrizenrechnung (Matrizenkalkül) ohne Computereinsatz ist nicht sinnvoll.

Lies dazu: http://www.mathematik.de/ger/information/kalenderblatt/cayleyundeisenstein/cayleyundeisenstein.html?print=1 (7.5.2011)

```
(%i19) x:invert(A).b;
```

(%o19) $\begin{bmatrix} -2 \\ 22 \\ 8 \end{bmatrix}$

```
(%i20) x:A^^(-1).b;
```

(%o20) $\begin{bmatrix} -2 \\ 22 \\ 8 \end{bmatrix}$

2.3 Determinantenmethode

Figure 7:

Die Matrix A lautete:	Der Vektor b lautete:	Die Lösung lautet:
-2, 1, -3, 7, -3, 10, 6, -2, 7.	2, 0, 0.	-2, 22, 8.

```
(%i21) A;
```

(%o21) $\begin{bmatrix} -2 & 1 & -3 \\ 7 & -3 & 10 \\ 6 & -2 & 7 \end{bmatrix}$

```
(%i22) b;
```

(%o22) $\begin{bmatrix} 2 \\ 0 \\ 0 \end{bmatrix}$

```
(%i23) A1: matrix(
 [2,1,-3],
 [0,-3,10],
 [0,-2,7]
)
/* die erste Spalte wurde gegen b getauscht */;
```

(%o23) $\begin{bmatrix} 2 & 1 & -3 \\ 0 & -3 & 10 \\ 0 & -2 & 7 \end{bmatrix}$

```
(%i24) A2: matrix(
 [-2,2,-3],
 [7,0,10],
 [6,0,7]
)
/* die zweite Spalte wurde gegen b getauscht */;
```

(%o24) $\begin{bmatrix} -2 & 2 & -3 \\ 7 & 0 & 10 \\ 6 & 0 & 7 \end{bmatrix}$

```
(%i25) A3: matrix(
        [-2,1,2],
        [7,-3,0],
        [6,-2,0]
       )
       /* die dritte Spalte wurde gegen c getauscht */;
```

(%o25) $\begin{bmatrix} -2 & 1 & 2 \\ 7 & -3 & 0 \\ 6 & -2 & 0 \end{bmatrix}$

```
(%i26) M:determinant(A);
       M1:determinant(A1);
       M2:determinant(A2);
       M3:determinant(A3)
       /* ohne Computer würde man die Regel von Sarrus verwenden
          http://de.wikipedia.org/wiki/Regel_von_Sarrus */;
```

(%o26) 1
(%o27) -2
(%o28) 22
(%o29) 8

```
(%i30) x:M1/M;y:M2/M;z:M3/M
       /* das ist die Berechnung der Lösung nach der Cramerschen Regel */;
```

(%o30) -2
(%o31) 22
(%o32) 8

3 Übungsaufgabe

Figure 8:



Die Matrix A lautete:
2, 1, 4,
13, 4, 17,
30, 2, 13.

Der Vektor b lautete:
4,
-1,
-2.

Die Lösung lautet:
75,
1422,
-392.

4 Übungsaufgabe

Figure 9:

Die Matrix A lautete:
3, 1, 3,
5, 3, 10,
26, 3, 5.

Der Vektor b lautete:
4,
-3,
0.

Die Lösung lautet:
-72,
1129,
-303.